ELECTRONOTES
1 PHEASANT LANE
ITHACA, NY 14850
(607)-273-8030

APPLICATION NOTE NO. 256
July 20, 1982

TI-59 PROGRAM FOR |T(s)| WITHOUT FACTORING

In AN-251, we have given a program for $|T(s)|$ based on knowledge of the poles and zeros of T(s). In many cases, T(s) is known as a ratio of polynomials such as:

$$T(s) = \frac{a_o + a_1 s + a_2 s^2 + \ldots\ldots + a_M s^M}{b_o + b_1 s + b_2 s^2 + \ldots\ldots + b_N s^N} = \frac{N(s)}{D(s)} \tag{1}$$

and we may not be given the roots of the numerator (the zeros) or the roots of the denominator (the poles). In such a case, it is possible to factor the polynominals (trivially if M or N is 2 or less) at least numerically. However, this is not necessary, as the frequency response can be obtained from the a_n and b_n already known. Here we will closely parallel the case used for H(z) in AN-255. Basically we will look at the frequency response as the magnitude of T(s), and this in turn as the ratio of the magnitude of the numerator to the magnitude of the denominator. To find these two magnitudes, we will separate out the real and imaginary parts of the polynomials for $s = j\omega$, and take the square root of the sum of the squares of the real part and the imaginary part.

First, substituting $j\omega$ for s, and looking just at N(s) we have:

$$\begin{aligned} N(j\omega) &= a_o + ja_1\omega + j^2a_2\omega^2 + j^3a_3\omega^3 + j^4a_4\omega^4 + \ldots\ldots j^M a_M \omega^M \\ &= a_o + ja_1\omega - a_2\omega^2 - ja_3\omega^3 + a_4\omega^4 + \ldots\ldots j^M a_M \omega^M \\ &= [a_o - a_2\omega^2 + a_4\omega^4 - a_6\omega^6 + \ldots\ldots] \\ &\quad +j[a_1\omega - a_3\omega^3 + a_5\omega^5 - a_7\omega^7 \ldots\ldots] \end{aligned} \tag{2}$$

From this we see that the separation of the real and imaginary parts is basically a multiplication of even and odd coefficients by appropriate powers of ω, and the proper summation allowing for alternating signs. To write a compact form for the summation, we can consider both M and N to be even. If this is not true, then the corresponding a_M and/or b_N can be set to zero for that particular case. Then for the numerator:

$$|N(j\omega)| = \left[[\sum_{m=0}^{M/2}(-1)^m a_{2m}\omega^{2m}]^2 + [\sum_{m=1}^{M/2}(-1)^{m-1} a_{2m-1}\omega^{2m-1}]^2 \right]^{\frac{1}{2}} \tag{3}$$

Handling the denominator in the same way, we get for T(s) as a whole:

$$|T(s)| = \left[\frac{\left[\sum_{m=0}^{M/2}(-1)^m a_{2m}\omega^{2m}\right]^2 + \left[\sum_{m=1}^{M/2}(-1)^{m-1} a_{2m-1}\omega^{2m-1}\right]^2}{\left[\sum_{n=0}^{N/2}(-1)^n b_{2n}\omega^{2n}\right]^2 + \left[\sum_{n=1}^{N/2}(-1)^{n-1} b_{2n-1}\omega^{2n-1}\right]^2} \right]^{\frac{1}{2}} \tag{4}$$

Equation (4) is a compact way of writing out the solution. However, in an actual program, it may be more useful to work with serially increasing values of a_n and b_n, doing first one term of one summation, and then doing a term of another summation, and so on, instead of doing the summations one at a time. The TI-59 program given here in fact does this sort of summation jumping.

Frequency Response Program for T(s)

```
1.  2nd Lbl A 2nd Cms STO 59  1 STO 50 RCL 59 R/S
2.  2nd Lbl B STO 58 R/S
3.  2nd Lbl 2nd A' 0 STO 57 R/S 2nd Lbl 2nd CP STO 2nd Ind 57 RCL 57 + 1 =
4.                     STO 57 R/S    GTO 2nd CP
5.  2nd Lbl 2nd B' 20 STO 56 R/S  2nd Lbl CE STO 2nd Ind 56  RCL 56 + 1 =
6.                     STO 56 R/S    GTO CE
7.  2nd Lbl E STO 50 R/S
8.  2nd Lbl 2nd C' STO 49 R/S
9.  2nd Lbl C ÷ RCL 49 =   GTO D
10. 2nd Lbl D STO 55 RCL 59 x⇄t  0 STO 54 STO 53 STO 52   1 STO 51
11.      2nd Lbl 2nd Eng RCL 55 y^x RCL 54 × RCL 2nd Ind 54 × RCL 51 = SUM 53
12.      RCL 54 x=t EE      + 1 = STO 54
13.                         RCL 55 y^x RCL 54 × RCL 2nd Ind 54 × RCL 51 = SUM 52
14.      RCL 54 x=t EE      + 1 = STO 54   RCL 51 +/- STO 51 GTO 2nd Eng
15.      2nd Lbl EE RCL 53 x^2 + (RCL 52 x^2) = √x STO 48
16.                    RCL 58 x⇄t 20 STO 54  0 STO 53 STO 52 1 STO 51
17.      2nd Lbl 2nd Fix RCL 55 y^x (RCL 54 - 20) × RCL 2nd Ind 54 × RCL 51 = SUM 53
18.      RCL 54 - 20 = x=t 2nd P→R  + 21 = STO 54
19.                         RCL 55 y^x (RCL 54 - 20) × RCL 2nd Ind 54 × RCL 51 = SUM 52
20.      RCL 54 - 20 = x=t 2nd P→R  + 21 = STO 54   RCL 51 +/- STO 51 GTO 2nd Fix
21.      2nd Lbl P→R RCL 53 x^2 + (RCL 52 x^2) = √x
22.         1/x × RCL 48 × RCL 50 =  R/S
```

Program Comments: Lines 1 & 2, set order. Lines 3&4 load numerator, 5&6 load denom. Lines 7&8 set initial multipliers. Line 9 normalizes freq. input. Lines 10-22 do freq. resp. calculation for a frequency stored in 55. 54 stores the exponent under consideration, 53 the real part, 52 the imaginary part, and 51 keeps track of the sign.

User Defined Keys

A' Start Entry of Numerator Coeff	B' Start Entry of Denom. Coeff.	C' Enter Normalizing Factor (optional)		
A Enter Order of Numerator	B Enter Order of Denominator	C Enter Frequencies to be Normalized	D Enter Frequencies Directly	E Enter Overall Multiplier

INSTRUCTIONS: (Note: instructions are nearly identical to AN-255, see for more information)

1. Enter order of numerator, Press A. Enter order of denominator, Press B. (both ≤ 20)
2. Press A', enter a_0, press R/S, enter a_1, press R/S, etc. (enter numerator)
3. Press B', enter b_0, press R/S, enter b_1, press R/S, etc. (enter denominator)
4. If frequency is to be normalized to any value, enter this using C', and then enter frequencies using the C key.
5. If an overall multiplier other than 1 is to be used, enter this with the E key. The E key can be used later as well if it is desired to normalize the frequency response at any frequency. [see instruction 7 of AN-255)
6. Enter frequencies at which you want to find the response. Frequencies may be entered in radians using the D key, or in arbitrary units using the C key. If frequencies are to be entered in Hertz, enter $1/2\pi$ with the C' key. It may take a bit of trial and error to find the features of the response unless something is known about the expected response (such as a cutoff frequency for example).